
Subject: Re: Mapping on Objekts...

Posted by [J.D. Smith](#) on Thu, 20 Mar 1997 08:00:00 GMT

[View Forum Message](#) <> [Reply to Message](#)

Hey Folks,

If you liked the cylinder, here's one that's even more fun (is it possible?!?).

Herr Moebius' Strip:

Put in a file and .run it. Same conditions as with cylinder apply.

Enjoy,

JD

```
:: BEGIN MOEBIUS STRIP CODE
```

```
scale3,xrange=[0,1],yrange=[0,1],zrange=[0,1],AZ=80,AX=50
N=125
phi=findgen(N)/(N-1)*2*!PI
;; define angle of strip section rotation
thet=phi/2
;; set up base radius of strip
Ro=.5
;; define half width of strip
d=.3
;; define the radius at the top and bottom (as defined at phi=0)
Rtop=Ro-d*sin(thet)      ;radius at top
Rbottom=Ro+d*sin(thet)   ;radius at bottom
zc=d*cos(thet)           ;height
;; convert polar to rectangular (top and bottom)
cv=cv_coord(FROM_POLAR=[rotate(phi,1),rotate(Rtop,1)],/TO_RECT)
xctop=cv(0,*) & yctop=cv(1,*)
cv=cv_coord(FROM_POLAR=[rotate(phi,1),rotate(Rbottom,1)],/TO_RECT)
xcbot=cv(0,*) & ycbot=cv(1,*)

;; get some example data
file=filepath('people.dat',subdir='examples/data')
openr,un,/get_lun,file
d=bytarr(192,192)
readu,un,d
free_lun,un
s=size(d)

set_plot,'Z'
erase
```

```
;; render Moebius in strips
for i=0,N-2 do begin
  x=[xcbot(i),xctop(i),xctop(i+1),xcbot(i+1)]+.5
  y=[ycbot(i),yctop(i),yctop(i+1),ycbot(i+1)]+.5
  z=[-zc(i),zc(i),zc(i+1),-zc(i+1)]+.5
  left=round(float(i)/N*(s(1)-1)) & right=round(float(i+1)/N*(s(1)-1))
  top=s(2)-1 & bot=0
  polyfill,x,y,z,/T3D,PATTERN=d(left:right,bot:top),/IMAGE_INT ERP, $
  IMAGE_COORD=[[0,0],[0,top],[right-left,top],[right-left,0]]
endfor
im=tvrd()
set_plot,'X'
tv,im
end
```
